


CHASE puts you in a maze made up of high-voltage fences and posts. This in itself isn't too unpleasant but there're also the five interceptor robots bent on just one thing—your destruction. If these robots touch you ... that's the end of the game (and you!). There's one hope—make the robots hit the maze, or each other (they're like people—sometimes they'd rather be alone). If you destroy them all, you win! If you find yourself in a totally hopeless situation, you have the option of making a tremendous leap to a random location (which may well be on top of a fence or a guard).

At the end of the game, you may replay with the same or different starting conditions.

I believe this game was originally created by Mac Oglesby. It was then modified by Bill Cotter and further improved by Arnold Loveridge. An intermediate version appeared in *Creative Computing*, Jan/Feb 1976.



RUN

CHASE  
CREATIVE COMPUTING  
HORRISTOWN, NEW JERSEY

YOU ARE WITHIN THE WALLS OF A HIGH VOLTAGE MAZE  
THERE ARE FIVE SECURITY MACHINES TRYING TO DESTROY YOU  
YOU ARE THE '+' THE INTERCEPTORS ARE THE 'X'  
THE AREAS MARKED 'X' ARE HIGH VOLTAGE  
YOUR ONLY CHANCE FOR SURVIVAL IS TO MANEUVER EACH  
INTERCEPTOR INTO AN 'X'.-----GOOD LUCK-----  
MOVES ARE 7.8.9  
4.\*.6  
1.2.3

10 = NO MOVE FOR THE REST OF THE GAME  
-1 = GAVE UP, SITUATION HOPELESS.  
0 = A TREMENDOUS (BUT UNFORTUNATELY RANDOM) LEAP

```

XXXXXXXXXXXXXXXXXXXXX
X                X  XX
X X +          X X  X
X                X X  X
X                X  X
X X +          X  +X
X X  X X XXX  X
XX    XX +    X
X *  X        + X
XXXXXXXXXXXXXXXXXXXXX
? 5
XXXXXXXXXXXXXXXXXXXXX
X                X  XX
X X          X X  X
X +          X X  X
X                X  X
X X          X  X
X X  +X X XXX+ X
XX    XX      X
X +  X +    + X
XXXXXXXXXXXXXXXXXXXXX
? 5
XXXXXXXXXXXXXXXXXXXXX
X                X  XX
X X          X X  X
X                X X  X
X +          X  X
X X          X  X
X X  X X XXX  X
XX  + XX    + X
X *  X +    + X
XXXXXXXXXXXXXXXXXXXXX
? 5
XXXXXXXXXXXXXXXXXXXXX
X                X  XX
X X          X X  X
X                X X  X
X +          X  X
X X          X  X
X X  X X XXX  X
XX  + XX    + X
X *  X +    + X
XXXXXXXXXXXXXXXXXXXXX
? 4
XXXXXXXXXXXXXXXXXXXXX
X                X  XX
X X          X X  X
X                X X  X
X                X  X
X X          X  X
X X  X X XXX  X
XX    XX      X
X ++  X ++    X
XXXXXXXXXXXXXXXXXXXXX
? 0
$6,000,000 JUMP!!!
HIGH VOLTAGE!!!!!!!!!!
***** ZAP ***** YOU'RE DEAD!!!

ANOTHER GAME (Y/N)? N
    
```

```

LIST
10 PRINT TAB(26);"CHASE"
20 PRINT TAB(20);"CREATIVE COMPUTING"
30 PRINT TAB(18);"MORRISTOWN, NEW JERSEY"
40 PRINT:PRINT:PRINT
41 PRINT "YOU ARE WITHIN THE WALLS OF A HIGH VOLTAGE MAZE"
42 PRINT "THERE ARE FIVE SECURITY MACHINES TRYING TO DESTROY YOU"
60 PRINT "YOU ARE THE '+' THE INTERCEPTORS ARE THE '-'"
70 PRINT "THE AREAS MARKED 'X' ARE HIGH VOLTAGE"
80 PRINT "YOUR ONLY CHANCE FOR SURVIVAL IS TO MANEUVER EACH"
90 PRINT "INTERCEPTOR INTO AN 'X'.-----GOOD LUCK-----"
100 PRINT "MOVES ARE 7.8.9"
110 PRINT "      4.*.6"
120 PRINT "      1.2.3"
130 PRINT
140 PRINT "10 = NO MOVE FOR THE REST OF THE GAME"
150 PRINT "-1 = GAVE UP, SITUATION HOPELESS."
160 PRINT " 0 = A TREMENDOUS (BUT UNFORTUNATELY RANDOM) LEAP"
170 PRINT
180 DIM A(10,20),A1(10,20),N(12),L(5),M(5),L1(5),M1(5)
190 REM
210 FOR B=1 TO 10
220 FOR C=1 TO 20
230 X=INT(10*RND(1))
240 IF X=5 THEN 270
250 A(B,C)=ASC(" ")
260 GOTO 280
270 A(B,C)=ASC("X")
280 NEXT C
290 NEXT B
300 FOR D=1 TO 10
310 A(D,1)=ASC("X"):A(D,20)=ASC("X")
320 NEXT D
330 FOR F=1 TO 20
340 A(1,F)=ASC("X"):A(10,F)=ASC("X")
350 NEXT F
360 GOTO 410
370 H=INT(2+8*RND(1))
380 I=INT(2+18*RND(1))
390 IF A(H,I) <> ASC("-") THEN 370
400 RETURN
410 GOSUB 370
420 A(H,I)=ASC("+")
430 J=H:K=I
440 FOR N9=1 TO 5
450 GOSUB 370
460 A(H,I)=ASC("+")
470 L(N9)=H:M(N9)=I
480 NEXT N9
490 FOR B1=1 TO 10:FOR B2=1 TO 20:A1(B1,B2)=A(B1,B2):NEXT B2:NEXT B1
500 FOR B1=1 TO 5:L1(B1)=L(B1):M1(B1)=M(B1):NEXT B1
510 J1=J:K1=K
520 Y9=0
530 FOR D2=1 TO 10
540 FOR B2=1 TO 20
550 N%=CHR$(A(D2,B2))
560 PRINT N%;
570 NEXT B2
580 PRINT
590 PRINT
600 NEXT D2
610 IF Y9 <> 10 THEN 640
620 PRINT
630 GOTO 890
640 INPUT Y9
650 J2=J:K2=K
660 IF Y9=0 THEN 860
670 IF Y9 < 0 THEN 1230
680 IF Y9=10 THEN 1070
690 ON Y9 GOTO 820,800,780,840,890,760,700,720,740
700 J=J-1:K=K-1
710 GOTO 890
720 J=J-1
730 GOTO 890
740 J=J-1:K=K+1
750 GOTO 890
760 K=K+1
770 GOTO 890
780 J=J+1:K=K+1
790 GOTO 890
800 J=J+1
810 GOTO 890
820 J=J+1:K=K-1
830 GOTO 890
840 K=K-1
850 GOTO 890
860 PRINT "86,000,000 JUMP!!!"
870 J=INT(2+8*RND(1))
880 K=INT(2+18*RND(1))
890 IF A(J,K)=ASC("X") THEN 1260
900 A(J2,K2)=ASC(" ")
910 A(J,K)=ASC("-")
920 GOTO 1070
930 REM INTERCEPTOR MOVEMENT
940 IF A(X,Y)=ASC("X") THEN 1040
950 X2=X:Y2=Y
960 X=SGN(J-X):Y=SGN(K-Y)
970 X=X+X2:Y=Y+Y2
980 IF A(X,Y)=ASC("-") THEN 1050
990 IF A(X,Y)=ASC(" ") THEN 1020
1000 A(X2,Y2)=ASC(" ")
1010 RETURN
1020 A(X,Y)=ASC("+")
1030 A(X2,Y2)=ASC(" ")
1040 RETURN
1050 G9=99
1060 RETURN
1070 FOR N9=1 TO 5
1080 X=L(N9):Y=M(N9)
1090 G9=0
1100 GOSUB 940
1110 IF G9 <> 0 THEN 1240
1120 L(N9)=X:M(N9)=Y
1130 NEXT N9
1140 FOR N9=1 TO 5
1150 IF A(L(N9),M(N9)) <> ASC(" ") THEN 1170
1160 A(L(N9),M(N9))=ASC("+")
1170 NEXT N9
1180 FOR N9=1 TO 5
1190 IF A(L(N9),M(N9)) <> ASC("X") THEN 540
1200 NEXT N9
1210 PRINT "YOU HAVE DESTROYED ALL YOUR OPPONENTS - THE GAME IS YOURS"
1220 GOTO 1290
1230 PRINT "GIVE UP, EH."
1240 PRINT "*** YOU HAVE BEEN DESTROYED BY A LUCKY COMPUTER ***"
1250 GOTO 1290
1260 PRINT "HIGH VOLTAGE!!!!!!!!!!!!!"
1270 PRINT "***** ZAP ***** YOU'RE DEAD!!!"
1280 PRINT
1290 PRINT "ANOTHER GAME (Y/N)";
1300 INPUT N9%
1310 IF N9% <> "Y" THEN 1400
1320 PRINT "SAME SETUP (Y/N)";
1330 INPUT N9%
1340 IF N9% <> "Y" THEN 190
1350 FOR B1=1 TO 10:FOR B2=1 TO 20:A(B1,B2)=A1(B1,B2):NEXT B2:NEXT B1
1360 FOR B1=1 TO 5:L1(B1)=L1(B1):M1(B1)=M1(B1):NEXT B1
1380 J=J1:K=K1
1390 GOTO 530
1400 END
OR

```

# Roadrace

You are the driver of a race car on the notorious NY Route 20. You'll have to drive 5 miles with  $\frac{1}{2}$  gallon of gas, while keeping alert for changes in the road conditions, other cars, etc.


At the start you pick your car and course. During the race you control braking and acceleration.

Watch out for passing another car! If you try to go the same speed he's going, you're going to meet a Greyhound bus head-on!

The game is tough to win. I usually wipe out in a curve or run out of gas. You might want to increase your MPG rating...look at line 870.

Good luck!

This program originally appeared in *Creative Computing*, Jan/Feb 1975.



RUN  
ROADRACE  
CREATIVE COMPUTING MORRISTOWN NEW JERSEY

THIS IS THE PITTSFIELD-ALBANY ROAD RALLY

WELCOME TO THE FIRST ANNUAL PITTSFIELD-ALBANY ROAD RALLY.  
YOU'LL BE DRIVING RT. 20. TRYING TO WIN THE RACE AND  
STAY ALIVE IN THE BARGIN. GOOD LUCK!!

YOU HAVE YOUR CHOICE OF: (1) A VW; (2) 283 NOVA;  
(3) Z-28; OR (4) FERRARI

CHOOSE THE CAR YOU WANT BY THE NUMBER IN FRONT OF IT.  
REMEMBER, THE BETTER THE CAR, THE MORE GAS IT USES.  
WHICH CAR? 3

NOW YOU CHOOSE WHICH COURSE YOU WANT TO RACE ON.  
THE EASIEST COURSE IS NUMBER 1, AND IS THE STRAIGHTEST  
ROUTE. NUMBER 3 CONSISTS MOSTLY OF TURNS AND TWISTS.  
WHICH ROUTE DO YOU WANT? 1

YOU WILL NEED TO TRAVEL 5 MILES WITH .5 GALLONS OF GAS.  
YOUR STATUS WILL BE SHOWN EACH 10 SECONDS. AFTER EACH  
STATUS CHECK YOU WILL BE ASKED FOR A NEW RATE OF GAS.  
A RATE OF +10 IS HARD ACCELERATION, AND -10 IS HARD BRAKING  
ANY NUMBER IN BETWEEN IS ALLOWABLE.

PRESENT VELOCITY = 0 NO. OF GALLONS = .5  
NO. OF MILES = 0 TIME PASSED = 0 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 10

ROAD CONDITIONS: CLEAR AND STRAIGHT

PRESENT VELOCITY = 70 NO. OF GALLONS = .47  
NO. OF MILES = .152174 TIME PASSED = 10 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 2

ROAD CONDITIONS: VEHICLE AHEAD 1000 FEET

PRESENT VELOCITY = 55 NO. OF GALLONS = .464  
NO. OF MILES = .271739 TIME PASSED = 20 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 10

ROAD CONDITIONS: VEHICLE PASSED BY 75 MPH

PRESENT VELOCITY = 102 NO. OF GALLONS = .434  
NO. OF MILES = .493478 TIME PASSED = 30 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 10

ROAD CONDITIONS: VEHICLE AHEAD 1000 FEET

PRESENT VELOCITY = 130 NO. OF GALLONS = .404  
NO. OF MILES = .776087 TIME PASSED = 40 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? -5

ROAD CONDITIONS: VEHICLE BEING PASSED  
GREYHOUND BUS IN OTHER LANE DOING 64 MPH CRASH VELOCITY = 105  
WHERE IS YOUR FUNERAL BEING HELD ?

YOU WANT TO TRY AGAIN, RIGHT !!!!  
1-YES, 2-NO? 1  
WHICH CAR? 3

WHICH ROUTE DO YOU WANT? 5

PRESENT VELOCITY = 0 NO. OF GALLONS = .5  
NO. OF MILES = 0 TIME PASSED = 0 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 10

ROAD CONDITIONS: WARNING: CURVE AHEAD

PRESENT VELOCITY = 70 NO. OF GALLONS = .47  
NO. OF MILES = .152174 TIME PASSED = 10 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 0

ROAD CONDITIONS: THROUGH CURVE

PRESENT VELOCITY = 41 NO. OF GALLONS = .47  
NO. OF MILES = .241304 TIME PASSED = 20 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 7

ROAD CONDITIONS: WARNING: CURVE AHEAD

PRESENT VELOCITY = 73 NO. OF GALLONS = .449  
NO. OF MILES = .4 TIME PASSED = 30 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? -1

ROAD CONDITIONS: THROUGH CURVE

PRESENT VELOCITY = 36 NO. OF GALLONS = .449  
NO. OF MILES = .478261 TIME PASSED = 40 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 10

ROAD CONDITIONS: CLEAR AND STRAIGHT

PRESENT VELOCITY = 91 NO. OF GALLONS = .419  
NO. OF MILES = .676087 TIME PASSED = 50 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 5

ROAD CONDITIONS: WARNING: CURVE AHEAD

PRESENT VELOCITY = 88 NO. OF GALLONS = .404  
NO. OF MILES = .867391 TIME PASSED = 60 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? -4

ROAD CONDITIONS: THROUGH CURVE

PRESENT VELOCITY = 23 NO. OF GALLONS = .404  
NO. OF MILES = .917391 TIME PASSED = 70 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? 7

ROAD CONDITIONS: WARNING: CURVE AHEAD

PRESENT VELOCITY = 62 NO. OF GALLONS = .383  
NO. OF MILES = 1.05217 TIME PASSED = 80 SECONDS  
WHAT IS YOUR NEW RATE OF GAS ? -1

ROAD CONDITIONS: ARE TERRIBLE  
16 WAS THE SPEED THROUGH THE CURVE  
29 WAS YOUR SPEED, BY THE WAY WHERE IS YOUR FUNERAL BEING HELD ?

```

10 PRINT TAB(27);"ROADRACE"
20 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN NEW JERSEY"
30 PRINT
40 PRINT
50 PRINT
100 PRINT " THIS IS THE PITTSFIELD-ALBANY ROAD RALLY"
120 PRINT
130 PRINT "WELCOME TO THE FIRST ANNUAL PITTSFIELD-ALBANY ROAD RALLY."
140 PRINT "YOU'LL BE DRIVING RT. 20, TRYING TO WIN THE RACE AND"
150 PRINT "STAY ALIVE IN THE BARGIN. GOOD LUCK!!"
160 PRINT
170 PRINT "YOU HAVE YOUR CHOICE OF: (1) A VW; (2) 283 NOVA;"
180 PRINT "(3) Z-28; OR (4) FERRARI"
190 PRINT
200 PRINT "CHOOSE THE CAR YOU WANT BY THE NUMBER IN FRONT OF IT."
210 PRINT "REMEMBER, THE BETTER THE CAR, THE MORE GAS IT USES."
220 PRINT "WHICH CAR?";
230 INPUT C1
240 LET C1=INT(C1)
250 IF C1 > 4 THEN 280
260 IF C1 < 1 THEN 280
270 GOTO 300
280 PRINT "INVALID CAR NUMBER. NEW CAR ";
290 GOTO 230
300 PRINT
310 IF N2=1 THEN 345
320 PRINT "NOW YOU CHOOSE WHICH COURSE YOU WANT TO RACE ON."
330 PRINT "THE EASIEST COURSE IS NUMBER 1, AND IS THE STRAIGHTEST"
340 PRINT "ROUTE. NUMBER 5 CONSISTS MOSTLY OF TURNS AND TWISTS."
345 PRINT "WHICH ROUTE DO YOU WANT?";
350 INPUT C2
360 LET C2=INT(C2)
380 IF C2 < 1 THEN 410
390 IF C2 > 5 THEN 410
400 GOTO 430
410 PRINT "INVALID COURSE NUMBER. NEW CHOICE ";
420 GOTO 350
430 IF N2=1 THEN 490
435 PRINT
440 PRINT "YOU WILL NEED TO TRAVEL 5 MILES WITH .5 GALLONS OF GAS."
450 PRINT "YOUR STATUS WILL BE SHOWN EACH 10 SECONDS. AFTER EACH "
460 PRINT "STATUS CHECK YOU WILL BE ASKED FOR A NEW RATE OF GAS. "
470 PRINT "A RATE OF +10 IS HARD ACCELERATION, AND -10 IS HARD BRAKING"
480 PRINT "ANY NUMBER IN BETWEEN IS ALLOWABLE."
490 FOR I=1 TO C1
500 READ B,K,S
510 LET B=B/10
520 NEXT I
530 LET A1=.5
540 LET M1=0
550 LET C1=C1/2
560 LET V=0
570 PRINT
580 LET R1=0
590 LET T=0
600 LET D=0
610 LET Q1=0
620 PRINT "PRESENT VELOCITY =";V;" NO. OF GALLONS =";A1
630 PRINT "NO. OF MILES =";M1;" TIME PASSED =";T;"SECONDS"
640 IF M1>= 5 THEN 1460
650 PRINT "WHAT IS YOUR NEW RATE OF GAS ";
660 INPUT G
670 IF G < -10 THEN 700
680 IF G > 10 THEN 700
690 GOTO 720
700 PRINT "NOT VALID. NEW RATE ";
710 GOTO 660
720 IF G < 9 THEN 780
730 LET Z=Z+1
740 IF Z > 4 THEN 760
750 GOTO 790
760 PRINT "YOUR ENGINE BLEW. YOU GOT HIT BY A PISTON."
770 GOTO 1270
780 LET Z=0
790 LET V=INT(B+G-K+V+V)
800 LET T=T+10
810 PRINT
820 PRINT "ROAD CONDITIONS:";
830 IF V > 0 THEN 850
840 LET V=0
850 LET M1=M1+V/460
860 IF G<0 THEN 890
870 LET A1=A1-(G*S)/5000

```

```

880 IF A1 < 0 THEN 1380
890 IF R1=1 THEN 1050
900 IF Q1=1 THEN 980
910 LET Q=INT((C2+1)*RND(1))
920 LET R=INT((3.75-C2)*RND(1))
930 IF R > 0 THEN 1290
940 IF Q > 0 THEN 1340
950 PRINT "CLEAR AN STRAIGHT"
960 PRINT
970 GOTO 620
980 LET H=INT(15+35!*RND(1))
990 LET H=H+5*C1
1000 IF V>H THEN 1500
1010 PRINT "THROUGH CURVE"
1020 PRINT
1030 LET Q1=0
1040 GOTO 620
1050 LET E=E-(V-D)*3!
1060 IF E < 0 THEN 1100
1070 PRINT "VECHICLE ";E;" FEET AHEAD"
1080 PRINT
1090 GOTO 620
1100 IF V-D < 5 THEN 1180
1110 PRINT "VECHICLE PASSED BY ";
1120 LET D=V-D
1130 PRINT D;
1140 PRINT "MPH"
1150 PRINT
1160 LET R1=0
1170 GOTO 620
1180 PRINT "VECHICLE BEING PASSED "
1190 LET D=INT(25+40*RND(1))
1200 PRINT "GREYHOUND BUS IN OTHER LANE ";
1210 PRINT "DOING ";
1220 PRINT D;
1230 PRINT " MPH ";
1240 LET D=V+D
1250 PRINT "CRASH VELOCITY = ";D
1270 PRINT "WHERE IS YOUR FUNERAL BEING HELD ?"
1280 GOTO 1560
1290 PRINT "VECHICLE AHEAD 1000 FEET"
1300 PRINT
1310 LET D=INT(25+35*RND(1))
1320 LET R1=1
1330 GOTO 620
1340 PRINT " WARNING: CURVE AHEAD "
1350 LET Q1=1
1360 PRINT
1370 GOTO 620
1380 PRINT "EXCELLENT BUT WAIT!"
1390 PRINT
1400 PRINT "YOU RAN OUT OF GAS"
1410 GOTO 1550
1420 PRINT "BUT SOME HOW YOU MADE IT"
1430 PRINT
1440 LET R1=0
1450 GOTO 620
1460 PRINT
1470 PRINT
1480 PRINT "YOU MADE IT (LUCKY) !!!!!!!"
1490 GOTO 1560
1500 PRINT "ARE TERRIBLE"
1510 LET H=H-5*C1
1520 PRINT H;" WAS THE SPEED THROUGH THE CURVE"
1530 PRINT V;" WAS YOUR SPEED, BY THE WAY ";
1540 GOTO 1270
1550 PRINT "YOU LEAD FOOTED $X3' '293333((3339'9(9((92'Z#Z#XZ"
1560 PRINT "YOU WANT TO TRY AGAIN, RIGHT !!!!!"
1570 PRINT "1-YES, 2-NO";
1580 INPUT V
1590 IF V=2 THEN 1620
1600 N2=1
1610 GOTO 1640
1620 PRINT "CHICKEN"
1630 GOTO 1700
1640 RESTORE
1650 GOTO 220
1660 DATA 45,.53,10
1665 DATA 60,.5,13
1670 DATA 70,-.41,15
1680 DATA 80,-.39,18
1700 END
OK

```